Slides 7 - Optimal Risky Portfolios

1. a. Diversification is _____. It _____.

   - ○ buying more than three stocks; reduces overall portfolio risk
   - ○ buying more than three stocks; increases the expected return
   - ○ the mixing of different assets within a portfolio; increases the expected return
   - ◉ the mixing of different assets within a portfolio; reduces overall portfolio risk

   Diversification is the mixing of different assets within a portfolio. It reduces overall portfolio risk

2. a. As different securities are added to a portfolio, the portfolio's total risk _____.

   - ○ increases
   - ○ is unaffected
   - ◉ decreases
   - ○ falls to zero

   The benefit of diversification is that total risk falls as more and more different securities are added to a portfolio, up to the point where all unsystematic risk is eliminated and only systematic risk remains.

3. a. Systematic risk affects _____.

   - ○ no firm if it is well diversified
   - ○ only individual firms
   - ○ only firms in a single industry
   - ◉ all firms

   Systematic risk is due to common risk factors, such as the state of the economy (the business cycle) or inflation and interest rates, which affect all firms in the market.

4. a. Unsystematic risk _____.

   Check all that apply:

   - [x] affects only a single asset or small group of assets
   - [x] can be diversified away
   - [ ] is measured by standard deviation
   - [x] is eliminated in a well-diversified portfolio

   Standard deviation measures total risk, while unsystematic risk is often measured as the standard deviation of the residuals.

5. The table below shows the expected rates of return for three stocks and their weights in some portfolio:

		Stock A	Stock B	Stock C
	Portfolio weights	0.4	0.2	0.4
State	Probability		Expected returns	
Recession	0.3	0.09	0.03	0.15
Boom	0.7	0.14	0.05	0.16

a. What is the portfolio return during a recession?

		Stock A	Stock B	Stock C	Portfolio
	Portfolio weights	0.4	0.2	0.4	
State	Probability		Expected returns		
Recession	0.3	0.09	0.03	0.15	0.102
Boom	0.7	0.14	0.05	0.16	0.13

The portfolio return is the weighted average of the individual expected returns:

$E(r_{P,\text{ recession}}) = w_A\, E(r_{A,\text{ recession}}) + w_B\, E(r_{B,\text{ recession}}) + w_C\, E(r_{C,\text{ recession}})$
= 0.4 * 0.09 + 0.2 * 0.03 + 0.4 * 0.15
= **0.102**

b. What is the expected portfolio return?

The expected portfolio return is the weighted average of the portfolio returns during a recession and a boom:

$E(r_P) = p_{recession}\, E(r_{P,\text{ recession}}) + p_{boom}\, E(r_{P,\text{ boom}})$

= 0.3 * 0.102 + 0.7 * 0.13

= **0.1216**

c. What is the standard deviation of the portfolio returns?

Variance of portfolio returns:

`sigma_P^2= sum_s p_s (E(r_(P,s))-E(r_P))^2`

= $0.3\,(0.102-0.1216)^2 + 0.7\,(0.13-0.1216)^2$

= 0.00016464

Standard deviation:

`σ = sqrt(σ^2) = 0.00016464^0.5=` **0.012831**

6. The return on Samsung stock has a standard deviation of 39% and the return on Toyota stock has a standard deviation of 24%. Their covariance is 0.03744.

a. If you invest 50% in Samsung and 50% in Toyota, what is the variance of the portfolio?

$\sigma^2 = w_S^2\, \sigma_S^2 + w_T^2\, \sigma_T^2 + 2\, w_S\, w_T\, cov(r_S, r_T)$

= $0.5^2 * 0.39^2 + 0.5^2 * 0.24^2 + 2 * 0.5 * 0.5 * 0.03744$

= **0.07115**

b. What is the standard deviation of the portfolio?

`σ = sqrt(σ^2) = 0.07115^0.5=` **0.2667**

7. IBM stock had a return of 14% two years ago, and 20% last year. The market portfolio had a return of 11% two years ago and 18% last year.

a. What was the average return on IBM stock?

`bar r_I = (0.14 + 0.2) / 2 =` **0.17**

b. What was the average return on the market portfolio?

`bar r_M = (0.11 + 0.18) / 2 =` **0.145**

c. What is the expected covariance of returns based on the historical data?

`cov = 1/(n-1) * [ (r_(I,1)-bar r_I)(r_(M,1)-bar r_M) + (r_(I,2)-bar r_I)(r_(M,2)-bar r_M)]`

= `1/(2-1) * [ (0.14 - 0.17)(0.11 - 0.145) + (0.2 - 0.17)(0.18 - 0.145)] `

= **0.0021**

8. The following table shows realized rates of return for two stocks.

	A	B	C
1	Year	Stock A	Stock B
2	1	-10%	15%
3	2	-15%	-14%
4	3	-6%	-5%
5	4	5%	28%
6	5	14%	8%
7	6	13%	4%

a. What is the arithmetic average return for stock B?

	A	B	C	D
1	Year	Stock A	Stock B	
2	1	-10%	15%	
3	2	-15%	-14%	
4	3	-6%	-5%	
5	4	5%	28%	
6	5	14%	8%	
7	6	13%	4%	
8	Average	0.0016667	**0.06**	=AVERAGE(C2:C7)

b. What is the variance for stock B?

	A	B	C	D
...				
9	Variance	0.015017	**0.02188**	=VAR.S(C2:C7)

c. What is the covariance of returns?

	A	B	C	D
...				
10	Covariance	**0.00776**		=COVARIANCE.S(B2:B7,C2:C7)

d. What is the correlation coefficient?

	A	B	C	D
...				
11	Correlation	**0.4281**		=CORREL(B2:B7,C2:C7)

9. The following table shows historical beginning-of-year prices for two stocks.

	A	B	C
1	Year	Stock A	Stock B
2	2015	36.06	576.93
3	2016	34.13	570.68
4	2017	33.5	592.61
5	2018	36.27	615.8
6	2019	33.97	540.85
7	2020	35.74	527.44
8	2021	30.79	486.12
9	2022	30.82	584.14
10	2023	31.7	637.52

a. Calculate the annual returns. What was the arithmetic average annual return for stock B?

	A	B	C	D
...				
12	Year	Stock A	Stock B	
13	2015	-0.05352	-0.010833	=C3/C2-1
14	2016	-0.018459	0.03843	
15	2017	0.08269	0.03913	
16	2018	-0.06341	-0.12171	
17	2019	0.0521	-0.02479	
18	2020	-0.1385	-0.07834	
19	2021	0.0009743	0.2016	
20	2022	0.02855	0.09138	
21				
22	Average	-0.013697	**0.016862**	=AVERAGE(C13:C20)

b. What was the variance for stock B?

	A	B	C	D
...				
23	Variance	0.005039	**0.010204**	=VAR.S(C13:C20)

c. What was the standard deviation for stock B?

	A	B	C	D
...				
24	St. Dev.	0.07099	**0.10101**	=STDEV.S(C13:C20)

d. What was the covariance of returns?

	A	B	C	D
...				

25	Covar.	**0.003577**		=COVARIANCE.S(B13:B20,C13:C20)

e. What was the correlation coefficient?

	A	B	C	D
...				
26	Correl.	**0.4988**		=CORREL(B13:B20,C13:C20)

10. The following table shows realized rates of return for two stocks.

	A	B	C
1	Year	Stock A	Stock B
2	1	5%	15%
3	2	-20%	-14%
4	3	-6%	-6%
5	4	5%	28%
6	5	14%	8%
7	6	9%	7%

a. What is the arithmetic average return for stock B?

	A	B	C	D
1	Year	Stock A	Stock B	
2	1	5%	15%	
3	2	-20%	-14%	
4	3	-6%	-6%	
5	4	5%	28%	
6	5	14%	8%	
7	6	9%	7%	
8	Average	0.011667	**0.06333**	=AVERAGE(C2:C7)

b. What is the standard deviation for stock B?

	A	B	C	D
...				
9	Variance	0.015097	0.02227	=VAR.S(C2:C7)
10	St. Dev.	0.12287	**0.14922**	=STDEV.S(C2:C7)

c. What is the correlation of returns?

	A	B	C	D
...				
11	Covariance	0.013233		=COVARIANCE.S(B2:B7,C2:C7)
12	Correlation	**0.7218**		=CORREL(B2:B7,C2:C7)

d. What is the expected return of a portfolio with 50% invested in stock A and the remainder in stock B?

	A	B	C	D

...				
14	Portfolio weight on A	Portfolio mean return	Portfolio standard deviation	
15	0.5	**0.0375**		=A15*B$8+(1-A15)*C$8

e. What is the standard deviation of a portfolio with 50% invested in stock A and the remainder in stock B?

	A	B	C	D
...				
14	Portfolio weight on A	Portfolio mean return	Portfolio standard deviation	
15	0.5	0.0375	**0.12632**	=SQRT(A15^2*B$9+(1-A15)^2 *C$9+2*A15*(1-A15)*B$11)

11. Go to Yahoo Finance (https://finance.yahoo.com/) and download monthly stock price data for Meta and Snap for all of 2022.

    a. Use the adjusted close prices to calculate monthly returns. What was the arithmetic average monthly return for Snap?

    The adjusted close prices are adjusted for dividend payments and stock splits. Therefore, it is likely that the adjusted close prices that you downloaded are different from the ones listed below. However, the returns will be the same.

	A	B	C	D
1	**Stock prices**			
2	Date	Meta	Snap	
3	1/1/22	313.26	32.54	
4	2/1/22	211.03	39.94	
5	3/1/22	222.36	35.99	
6	4/1/22	200.47	28.46	
7	5/1/22	193.64	14.11	
8	6/1/22	161.25	13.13	
9	7/1/22	159.1	9.88	
10	8/1/22	162.93	10.88	
11	9/1/22	135.68	9.82	
12	10/1/2	93.16	9.91	
13	11/1/22	118.1	10.31	
14	12/1/22	120.34	8.95	
15				
16	**Returns**			
17	Date	Meta	Snap	
18	1/1/22	-0.3263	0.2274	=C4/C3-1
19	2/1/22	0.05369	-0.0989	

20	3/1/22	-0.09844	-0.2092	
21	4/1/22	-0.03407	-0.5042	
22	5/1/22	-0.16727	-0.06945	
23	6/1/22	-0.013333	-0.2475	
24	7/1/22	0.02407	0.10121	
25	8/1/22	-0.16725	-0.09743	
26	9/1/22	-0.3134	0.009165	
27	10/1/22	0.2677	0.04036	
28	11/1/22	0.018967	-0.13191	
29	12/1/22			
30	Average	-0.0687	**-0.08914**	=AVERAGE(C18:C28)

b. What was the variance for Snap?

	A	B	C	D
...				
31	Variance	0.02953	**0.03771**	=VAR.S(C18:C28)

c. What was the standard deviation for Snap?

	A	B	C	D
...				
32	St.Dev.	0.17183	**0.19418**	=STDEV.S(C18:C28)

d. What was the covariance of returns?

	A	B	C	D
...				
33	Covar.	**-0.007002**		=COVARIANCE.S(B18:B28,C18:C28)

e. What was the correlation coefficient?

	A	B	C	D
...				
34	Correl.	**-0.2099**		=CORREL(B18:B28,C18:C28)

12. You found the following covariance matrix for 3 stocks:

	C	D	E	F
1		Amazon	Exxon	Walmart
2	Amazon	0.052	0.025	0.014
3	Exxon	0.025	0.044	0.0087
4	Walmart	0.014	0.0087	0.064

a. Add the border-multiplied covariance matrix, i.e., the covariance matrix bordered by the portfolio weights, below the table. What is the variance of the equally-weighted portfolio?

The border-multiplied covariance matrix is the matrix that is obtained by multiplying each covariance by the two corresponding weights in the border.

The weights in the vertical and horizontal border must be symmetric and linked. Enter =C8 in cell D7, =C9 in cell E7, and =C10 in cell F7. For an equally-weighted portfolio:

$w_A = 0.3333$
$w_E = 0.3333$
$w_W = 0.3333$

Portfolio variance:

`σ_P^2 = sum_(i=1)^3 sum_(j=1)^3 (w_iw_j cov(r_i, r_j))`

Example formula for cell D8:
= D$7 * $C8 * D2

The $ signs change the appropriate references from relative to absolute. The formula can thus be copied and pasted to the other 8 cells in the table without any manual adjustments.

	A	B	C	D	E	F
...						
6				Amazon	Exxon	Walmart
7			Weights	33.33%	33.33%	33.33%
8	Amazon		33.33%	0.005778	0.002778	0.0015556
9	Exxon		33.33%	0.002778	0.004889	0.0009667
10	Walmart		33.33%	0.0015556	0.0009667	0.007111
11				=D$7*$C10*D4		
12						
13	σ_P^2	**0.02838**	=SUM(D8:F10)			

b. You did extensive security research and came up with the following annual expected returns:

$E(r_A) = 0.07$
$E(r_E) = 0.06$
$E(r_W) = 0.05$

What is the expected return of the equally-weighted portfolio? Use Excel's SUMPRODUCT() function.

The portfolio expected return equals the weighted average expected return of the individual securities. The SUMPRODUCT() function is ideal for this purpose:

	A	B	C	D	E	F
...						
6				Amazon	Exxon	Walmart
7		E(r)	Weights	33.33%	33.33%	33.33%
8	Amazon	0.07	33.33%	0.005778	0.025	0.014
9	Exxon	0.06	33.33%	0.025	0.044	0.0087
10	Walmart	0.05	33.33%	0.014	0.0087	0.064
11				=D$7*$C10*D4		
12						
13	σ_P^2	0.02838	=SUM(D8:F10)			

14	$E(r_P)$	**0.06**	**=SUMPRODUCT(B8:B10, C8:C10)**	

c. What is the variance of the portfolio with the following weights?

$w_A = 0.4$
$w_E = 0.25$
$w_W = 0.35$

	A	B	C	D	E	F
...						
6				Amazon	Exxon	Walmart
7			Weights	40%	25%	35%
8	Amazon		40%	0.00832	0.0025	0.00196
9	Exxon		25%	0.0025	0.00275	0.0007613
10	Walmart		35%	0.00196	0.0007613	0.00784
11				=D$7*$C10*D4		
12						
13	σ_P^2	**0.02935**	=SUM(D8:F10)			

13. The following table shows realized rates of return for two stocks.

	A	B	C
1	Year	Stock A	Stock B
2	1	5%	15%
3	2	-16%	-14%
4	3	-6%	-5%
5	4	5%	28%
6	5	14%	8%
7	6	11%	4%

a. What is the arithmetic average return for stock B?

	A	B	C	D
1	Year	Stock A	Stock B	Formula
2	1	5%	15%	
3	2	-16%	-14%	
4	3	-6%	-5%	
5	4	5%	28%	
6	5	14%	8%	
7	6	11%	4%	
8	Average	0.02167	**0.06**	=AVERAGE(C2:C7)

b. What is the variance for stock B?

	A	B	C	D
...				
9	Variance	0.012617	**0.02188**	=VAR.S(C2:C7)

c. What is the covariance of returns?

	A	B	C	D
...				
10	Covariance	**0.01094**		=COVARIANCE.S(B2:B7,C2:C7)

d. What is the mean return of a portfolio with 20% invested in stock A and the remainder in stock B (assuming annual rebalancing)?

	A	B	C	D
...				
12	Portfolio weight on A	Portfolio mean return	Portfolio standard deviation	
13	0.2	**0.05233**		=A13*B$8+(1-A13)*C$8

e. What is the standard deviation of a portfolio with 20% invested in stock A and the remainder in stock B (assuming annual rebalancing)?

	A	B	C	D
...				
12	Portfolio weight on A	Portfolio mean return	Portfolio standard deviation	
13	0.2	0.05233	**0.1342**	=SQRT(A13^2*B$9+(1-A13)^2*C$9+2*A13*(1-A13)*B$10)

f. What is the mean return of a portfolio with 60% invested in stock A and the remainder in stock B (assuming annual rebalancing)?

	A	B	C	D
...				
12	Portfolio weight on A	Portfolio mean return	Portfolio standard deviation	
13	0.2	0.05233	0.1342	=SQRT(A13^2*B$9+(1-A13)^2*C$9+2*A13*(1-A13)*B$10)
14	0.6	**0.037**		=A14*B$8+(1-A14)*C$8

g. What is the standard deviation of a portfolio with 60% invested in stock A and the remainder in stock B (assuming annual rebalancing)?

	A	B	C	D
...				
12	Portfolio weight on A	Portfolio mean return	Portfolio standard deviation	
13	0.2	0.05233	0.1342	=SQRT(A13^2*B$9+(1-A13)^2*C$9+2*A13*(1-A13)*B$10)

14	0.6	0.037	**0.1153**	=SQRT(A14^2*B$9+(1-A14)^2*C$9+2*A14*(1-A14)*B$10)

14. Download this spreadsheet (problemData/CPG_prices.xlsx) with stock prices for five consumer packaged goods companies.

    a. What was the monthly return for Unilever (UN) from January to February 2015?

	A	B	C	...	F	G
...						
42	Returns					
43	Date	KO	MDLZ		UN	
44	1/1/15					
45	2/1/15	0.052	0.048		**0.00231**	=F3/F2-1

Make sure to show enough decimal places in your spreadsheet.

    b. What was the monthly return from November to December 2017 for a portfolio with the following weights?

	A	B	C	D	E	F
...						
39	Portfolio	KO	MDLZ	PEP	PG	UN
40	Weights	0.1	0.2	0.15	0.15	0.4

	A	B	C	...	F	G	H
...							
43	Date	KO	MDLZ		UN	Portfolio	
...							
79	12/1/17	0.011	-0.003		-0.017	**0.002051**	=SUMPRODUCT(B$40:F$40,B79:F79)

    c. What was the standard deviation of the same portfolio?

	A	B	C	...	F	G	H
...							
79	12/1/17	0.011	-0.003		-0.017	0.002051	=SUMPRODUCT(B$40:F$40,B79:F79)
80							
82	St. Dev.	0.03275	0.05054		0.05279	**0.03586**	=STDEV.S(G45:G79)

15. You have $5,000 to invest. You've done some security analysis and generated the following data for three stocks and Treasury bills, including weights in the optimal risky portfolio (ORP) from doing Markowitz portfolio optimization:

Security	Stock A	Stock B	Stock C	T-bills
Expected return (%)	12	10	5	4
Variance	0.04	0.03	0.02	0
Beta	1.2	1.5	0.8	0
Weight in ORP (%)	47	18	35	0

    a. What is the expected return of the optimal risky portfolio (ORP)?

Expected return of ORP:

`E(r_p) = w_A E(r_A) + w_B E(r_B) + w_C E(r_C)`

= `0.47 * 0.12 + 0.18 * 0.1 + 0.35 * 0.05`

= **0.0919**

b. How much money should you invest in the ORP to achieve an expected return of 8% for the complete portfolio (in $)?

Expected return of complete portfolio:

`E(r_"CP") = y E(r_p) + (1-y) r_f`

⇔ `y = (E(r_"CP") - r_f) / (E(r_p) - r_f)= (0.08 - 0.04) / (0.0919 - 0.04)= 0.7707`

You have to invest 77.07% of your $5,000 in the ORP:

`I_"ORP" = 0.7707 * 5,000 =` **3,854**

c. If you want to achieve an expected return of 8% for the complete portfolio, how much money should you invest in stock A (in $)?

Investment in stock A:

`I_A = I_"ORP" * w_A`

= `3,854 * 0.47`

= **1,811.18**

16. a. The optimal risky portfolio _____.

- ○ has no systematic risk
- ○ has the highest expected return
- ○ is the minimum-variance portfolio
- ◉ lies on the efficient frontier

The optimal risky portfolio is that portfolio on the efficient frontier that has the highest Sharpe ratio, i.e., the slope of the capital allocation line throught the risk-free asset and the portfolio is the steepest possible.

17. The return statistics for two stocks are given below:

	A	B	C
1		Stock A	Stock B
2	Expected return	0.12	0.072
3	Variance	0.0484	0.0324
4	Standard deviation	0.22	0.18
5	Covariance	0.02376	

a. What is the fraction of stock A in the minimum variance portfolio?

	A	B	C	D
1		Stock A	Stock B	
2	Expected return	0.12	0.072	
3	Variance	0.0484	0.0324	

4	Standard deviation	0.22	0.18	
5	Covariance	0.02376		
6				
7	Min. var. portfolio			
8	Investment weight	0.2596	0.7404	=(C3-B5)/(B3+C3-2*B5)

The fraction is **0.2596**.

b. What is the standard deviation of a portfolio with 30% invested in stock A and the rest in stock B?

	A	B	C	D	E
...					
10	Efficient frontier				
11	A	B	E(r)	SD	
12	0	1	0.072	0.18	=SQRT(A12^2*B$3+B12^2*C$3+2*A12*B12*B$5)
13	0.1	0.9	0.0768	0.17608	
14	0.2	0.8	0.0816	0.174	
15	0.3	0.7	0.0864	**0.17381**	=SQRT(A15^2*B$3+B15^2*C$3+2*A15*B15*B$5)

c. What is the standard deviation of a portfolio with 90% invested in stock A and the rest in stock B?

	A	B	C	D	E
...					
10	Efficient frontier				
11	A	B	E(r)	SD	
12	0	1	0.072	0.18	=SQRT(A12^2*B$3+B12^2*C$3+2*A12*B12*B$5)
13	0.1	0.9	0.0768	0.17608	
14	0.2	0.8	0.0816	0.174	
15	0.3	0.7	0.0864	0.17381	=SQRT(A15^2*B$3+B15^2*C$3+2*A15*B15*B$5)
16	0.4	0.6	0.0912	0.17554	
17	0.5	0.5	0.096	0.17911	
18	0.6	0.4	0.1008	0.18443	
19	0.7	0.3	0.1056	0.19134	
20	0.8	0.2	0.1104	0.19969	
21	0.9	0.1	0.1152	**0.2093**	=SQRT(A21^2*B$3+B21^2*C$3+2*A21*B21*B$5)

18. You have $70,000 to invest. You've done some security analysis and generated the following data for two stocks and Treasury bills:

Security	Stock A	Stock B	T-bills
Expected return (%)	12	9	2
Variance	0.04	0.0256	0
Correlation with stock A	1	0.5	0

a. What is the weight of stock A in the optimal risky portfolio (ORP)?

Adding the standard deviation for each stock:

Security	Stock A	Stock B	T-bills
Expected return (%)	12	9	2
Variance	0.04	0.0256	0
Standard deviation	0.2	0.16	0
Correlation with stock A	1	0.5	0

With only two risky assets, we can either do the full Markowitz portfolio optimization, or use the following formula to find the weights in the optimal risky portfolio:

Let ρ (rho) be the correlation coefficient:

`w_A = ((E(r_A)-r_f)σ_B^2-(E(r_B)-r_f) σ_A σ_B ρ) / ((E(r_A)-r_f)σ_B^2+(E(r_B)-r_f)σ_A^2 - (E(r_A)-r_f+E(r_B)-r_f)σ_Aσ_Bρ)`

= `((0.12-0.02) * 0.0256 - (0.09-0.02) * 0.2 * 0.16 * 0.5) / ((0.12-0.02) * 0.0256 + (0.09-0.02) * 0.04 - (0.12-0.02 + 0.09-0.02) * 0.2 * 0.16 * 0.5)`

= **0.5455**

b. If you invest 60% of your funds in T-Bills, what is the expected return of this complete portfolio?

Expected return of ORP:
$E(r_p) = w_A E(r_A) + w_B E(r_B)$
= 0.5455 * 0.12 + 0.4545 * 0.09
= 0.10636

Weight of ORP:
$w_P = 1 - w_f$
= 1 - 0.6
= 0.4

Expected return of complete portfolio:
$E(r_{CP}) = w_P E(r_p) + w_f r_f$
= 0.4 * 0.10636 + 0.6 * 0.02
= **0.05455**

c. What is the standard deviation of the optimal risky portfolio?

Variance of ORP:
$\sigma_P^2 = w_A^2 \sigma_A^2 + w_B^2 \sigma_B^2 + 2 w_A w_B \sigma_A \sigma_B \rho$
= $0.5455^2 * 0.04 + 0.4545^2 * 0.0256 + 2 * 0.5455 * 0.4545 * 0.2 * 0.16 * 0.5$
= 0.02512

Standard deviation of ORP:
$\sigma_P = (\sigma_P^2)^{0.5}$
= $0.02512^{0.5}$
= **0.15851**

d. What is the Sharpe ratio of your complete portfolio?

The Sharpe ratio of the complete portfolio is the same as the Sharpe ratio of the ORP, since they both lie on the same capital allocation line.

Sharpe ratio of ORP:

`S = (E(r_P) - r_f) / σ_P= (0.10636 - 0.02) / 0.15851=`**0.5449**

e. How much money do you have to invest in stock B to achieve this Sharpe ratio (in $)?

Investment in stock B:

$I_B = I * w_P * w_B$

= 70,000 * 0.4 * 0.4545

= **12,727**

19. The return statistics for two stocks and T-bills are given below:

	A	B	C	D
1		Stock A	Stock B	T-bills
2	Expected return	0.094	0.063	0.02
3	Variance	0.0961	0.0729	
4	Standard deviation	0.31	0.27	
5	Covariance	0.02511		

a. What is the Sharpe ratio of a portfolio with 70% invested in stock A and the rest in stock B?

	A	B	C	D
1		Stock A	Stock B	T-bills
2	Expected return	0.094	0.063	0.02
3	Variance	0.0961	0.0729	
4	Standard deviation	0.31	0.27	
5	Covariance	0.02511		
6				
7	**Portfolio**			
8	Weights	0.7	0.3	=1-B8
9	Expected return	0.0847		=B8*B2+C8*C2
10	Variance	0.0642		=B8^2*B3+C8^2*C3+2*B8*C8*B5
11	Standard deviation	0.2534		=B10^0.5
12	Sharpe ratio	**0.2554**		=(B9-D2)/B11

b. What is the Sharpe ratio of the optimal risky portfolio?

Use Excel's Solver tool to:

- Set objective: B12 (the Sharpe ratio)
- To: Max (maximize the Sharpe ratio)
- By changing variable cells: B8 (the weight on stock A)

	A	B	C	D
1		Stock A	Stock B	T-bills
2	Expected return	0.094	0.063	0.02
3	Variance	0.0961	0.0729	
4	Standard deviation	0.31	0.27	
5	Covariance	0.02511		
6				
7	**Portfolio**			
8	Weights	0.6549	0.3451	=1-B8

9	Expected return	0.0833		=B8*B2+C8*C2
10	Variance	0.06125		=B8^2*B3+C8^2*C3+2*B8*C8*B5
11	Standard deviation	0.2475		=B10^0.5
12	Sharpe ratio	**0.2558**		=(B9-D2)/B11

20. You found the following covariance matrix for 3 stocks:

	A	...	D	E	F
1			Amazon	Exxon	Walmart
2	Amazon		0.35	0.025	0.01
3	Exxon		0.025	0.21	0.02
4	Walmart		0.01	0.02	0.15

a. Add the border-multiplied covariance matrix, i.e., the covariance matrix bordered by the portfolio weights, below the table. What is the variance of the equally-weighted portfolio?

The border-multiplied covariance matrix is the matrix that is obtained by multiplying each covariance by the two corresponding weights in the border.

The weights in the vertical and horizontal border must be symmetric and linked. Enter =C8 in cell D7, =C9 in cell E7, and =C10 in cell F7. For an equally-weighted portfolio:

w_A = 0.3333
w_E = 0.3333
w_W = 0.3333

Portfolio variance:

`σ_P^2 = sum_(i=1)^3 sum_(j=1)^3 (w_iw_j cov(r_i, r_j))`

Example formula for cell D8:
= D$7 * $C8 * D2

The $ signs change the appropriate references from relative to absolute. The formula can thus be copied and pasted to the other 8 cells in the table without any manual adjustments.

	A	B	C	D	E	F
...						
6				Amazon	Exxon	Walmart
7			Weights	33.33%	33.33%	33.33%
8	Amazon		**33.33%**	0.03889	0.002778	0.0011111
9	Exxon		**33.33%**	0.002778	0.02333	0.002222
10	Walmart		**33.33%**	0.0011111	0.002222	0.016667
11				=D$7*$C10*D4		
12						
13	σ_P^2	**0.09111**	=SUM(D8:F10)			

b. You did extensive security research and came up with the following annual expected returns:

$E(r_A)$ = 0.07
$E(r_E)$ = 0.06
$E(r_W)$ = 0.05

What is the expected return of the equally-weighted portfolio? Use Excel's SUMPRODUCT() function.

The portfolio expected return equals the weighted average expected return of the individual securities. The SUMPRODUCT() function is ideal for this purpose:

	A	B	C	D	E	F
...						
6				Amazon	Exxon	Walmart
7		E(r)	Weights	33.33%	33.33%	33.33%
8	Amazon	0.07	**33.33%**	0.03889	0.002778	0.0011111
9	Exxon	0.06	**33.33%**	0.002778	0.02333	0.002222
10	Walmart	0.05	**33.33%**	0.0011111	0.002222	0.016667
11			100%	=D$7*$C10*D4		
12			=SUM(C8:C10)			
13	σ_P^2	0.09111	=SUM(D8:F10)			
14	$E(r_P)$	**0.06**	**=SUMPRODUCT(B8:B10, C8:C10)**			

c. The yield on Treasury bills is 1.1%. What is the Sharpe ratio of the equally-weighted portfolio?

Sharpe ratio:

`S = (E(r_P) - r_f) / sigma_P = (0.06-0.011)/0.09111^0.5 =`0.16233

	A	B	C	D	E	F
...						
6				Amazon	Exxon	Walmart
7		E(r)	Weights	33.33%	33.33%	33.33%
8	Amazon	0.07	**33.33%**	0.03889	0.002778	0.0011111
9	Exxon	0.06	**33.33%**	0.002778	0.02333	0.002222
10	Walmart	0.05	**33.33%**	0.0011111	0.002222	0.016667
11			100%	=D$7*$C10*D4		
12			=SUM(C8:C10)			
13	σ_P^2	0.09111	=SUM(D8:F10)			
14	$E(r_P)$	0.06	=SUMPRODUCT(B8:B10, C8:C10)			
15	r_f	0.011				
16	S	**0.16233**	**=(B14-B15)/B13^0.5**			

d. Determine the optimal risky portfolio (ORP) using Excel's Solver tool. What is the Sharpe ratio of the portfolio?

Using the Solver add-in for Excel or Google Sheets, we need to define the following:

- Set objective: B16
- To: Max
- By changing variable cells: C8:C10
- Subject to the constraints: C11=1

After running the tool by clicking on "Solve", we get the following solution:

	A	B	C	D	E	F
...						

6				Amazon	Exxon	Walmart
7		E(r)	Weights	0.2616	0.3428	0.3955
8	Amazon	0.07	**0.2616**	0.02395	0.002242	0.0010346
9	Exxon	0.06	**0.3428**	0.002242	0.02468	0.002712
10	Walmart	0.05	**0.3955**	0.0010346	0.002712	0.02346
11			100%	=D$7*$C10*D4		
12			=SUM(C8:C10)			
13	σ_P^2	0.08407	=SUM(D8:F10)			
14	$E(r_P)$	0.05866	=SUMPRODUCT(B8:B10, C8:C10)			
15	r_f	0.011				
16	S	**0.16436**	**=(B14-B15)**/**B13^0.5**			

The Sharpe ratio of the optimal risky portfolio is **0.16436**, a slight increase from the portfolio with equal weights.

21. You have $10,000 and want to invest it in the two stocks below and the risk-free asset, Treasury bills:

	A	B	C	D
1		Stock A	Stock B	T-bills
2	Expected return	0.091	0.061	0.02
3	Variance	0.1156	0.0729	
4	Standard deviation	0.34	0.27	
5	Covariance	0.02754		

a. What is the Sharpe ratio of the optimal risky portfolio?

Use Excel's Solver tool to:

- Set objective: B12 (the Sharpe ratio)
- To: Max (maximize the Sharpe ratio)
- By changing variable cells: B8 (the weight on stock A)

	A	B	C	D
1		Stock A	Stock B	T-bills
2	Expected return	0.091	0.061	0.02
3	Variance	0.1156	0.0729	
4	Standard deviation	0.34	0.27	
5	Covariance	0.02754		
6				
7	**Optimal risky portfolio**			
8	Weights	0.5924	0.4076	=1-B8
9	Expected return	0.07877		=B8*B2+C8*C2
10	Variance	0.06598		=B8^2*B3+C8^2*C3+2*B8*C8*B5
11	Standard deviation	0.2569		=B10^0.5
12	Sharpe ratio	**0.2288**		=(B9-D2)/B11

b. What is the standard deviation of a portfolio composed of $9,000 optimal risky portfolio and $1,000 risk-free asset?

	A	B	C	D
1		Stock A	Stock B	T-bills
2	Expected return	0.091	0.061	0.02
3	Variance	0.1156	0.0729	
4	Standard deviation	0.34	0.27	
5	Covariance	0.02754		
6				
7	**Optimal risky portfolio**			
8	Weights	0.5924	0.4076	=1-B8
9	Expected return	0.07877		=B8*B2+C8*C2
10	Variance	0.06598		=B8^2*B3+C8^2*C3+2*B8*C8*B5
11	Standard deviation	0.2569		=B10^0.5
12	Sharpe ratio	0.2288		=(B9-D2)/B11
13				
14	**Complete portfolio**			
15	Investment	10,000		
16	Inv. in ORP	9,000		
17	Weight on ORP	0.9		=B16/B15
18	Expected return	0.0729		=B17*B9+(1-B17)*D2
19	Standard deviation	**0.2312**		=B17*B11

c. Still assuming a portfolio composed of $9,000 optimal risky portfolio and $1,000 risk-free asset, how much money should you invest in stock B (in $)?

	A	B	C	D
1		Stock A	Stock B	T-bills
2	Expected return	0.091	0.061	0.02
3	Variance	0.1156	0.0729	
4	Standard deviation	0.34	0.27	
5	Covariance	0.02754		
6				
7	**Optimal risky portfolio**			
8	Weights	0.5924	0.4076	=1-B8
9	Expected return	0.07877		=B8*B2+C8*C2
10	Variance	0.06598		=B8^2*B3+C8^2*C3+2*B8*C8*B5
11	Standard deviation	0.2569		=B10^0.5
12	Sharpe ratio	0.2288		=(B9-D2)/B11
13				
14	**Complete portfolio**			
15	Investment	10,000		
16	Inv. in ORP	9,000		

17	Weight on ORP	0.9		=B16/B15
18	Expected return	0.0729		=B17*B9+(1-B17)*D2
19	Standard deviation	0.2312		=B17*B11
20	Inv. in stock B		**3,668**	=$B16*C8

22. The following table shows historical end-of-week adjusted close prices (including dividends) for two stocks.

	A	B	C
1	Week	Stock A	Stock B
2	0	39	2,748
3	1	39.64	2,690
4	2	42.57	2,731
5	3	41.94	2,830
6	4	39.79	2,883
7	5	42.09	2,809
8	6	43.86	2,884
9	7	39.64	2,823
10	8	39.97	2,921
11	9	40.85	3,074
12	10	42.02	3,037

a. Calculate the weekly returns. What is standard deviation of weekly returns for stock B?

	A	B	C	D
...				
14	Week	$r_{\text{Stock A}}$	$r_{\text{Stock B}}$	
15	1	0.01641	-0.02111	=C3/C2-1
16	2	0.07392	0.015242	
17	3	-0.014799	0.03625	
18	4	-0.05126	0.018728	
19	5	0.0578	-0.02567	
20	6	0.04205	0.0267	
21	7	-0.09622	-0.02115	
22	8	0.008325	0.03471	
23	9	0.02202	0.05238	
24	10	0.02864	-0.012036	=C12/C11-1
25	HPR	0.07744	0.10517	=C12/C2-1
26	Average	0.008689	0.010405	=AVERAGE(C15:C24)
27	Variance	0.002618	0.000797	=VAR.S(C15:C24)
28	Covariance	0.00012792		=COVARIANCE.S(B15:B24,C15:C24)
29	St. Dev.	0.05117	**0.02823**	=STDEV.S(C15:C24)

b. For all remaining parts, create a portfolio of 50% stock A and 50% stock B. If you rebalanced such a portfolio every week to keep the weights at 0.5/0.5, what is the holding period return over the 10 weeks for the portfolio?

	A	B	C	D
...				
31	Portfolio	Stock A	Stock B	
32	Weights	0.5	0.5	
33				
34	Rebalanced portfolio			
35	Week	r	1+r	
36	1	-0.002348	0.9977	=$B$32*B15+$C$32*C15
37	2	0.04458	1.0446	
38	3	0.010726	1.0107	
39	4	-0.016268	0.9837	
40	5	0.016068	1.0161	
41	6	0.03438	1.0344	
42	7	-0.05868	0.9413	
43	8	0.02152	1.0215	
44	9	0.0372	1.0372	
45	10	0.008302	1.0083	
46				
47	HPR	**0.09513**		=PRODUCT(C36:C45)-1

c. What is the standard deviation of weekly returns for such a portfolio if you rebalanced every week?

	A	B	C	D
...				
31	Portfolio	Stock A	Stock B	
32	Weights	0.5	0.5	
33				
34	Rebalanced portfolio			
35	Week	r	1+r	
36	1	-0.002348	0.9977	=$B$32*B15+$C$32*C15
37	2	0.04458	1.0446	
38	3	0.010726	1.0107	
39	4	-0.016268	0.9837	
40	5	0.016068	1.0161	
41	6	0.03438	1.0344	
42	7	-0.05868	0.9413	
43	8	0.02152	1.0215	
44	9	0.0372	1.0372	
45	10	0.008302	1.0083	
...				
48	Average	0.009547		=AVERAGE(B36:B45)
49	Variance	0.0009178		=VAR.S(B36:B45)

50	St. Dev.	0.0303		=STDEV.S(B36:B45)
51	Alternative	**0.0303**		=(B32^2*B27+C32^2*C27+2*B32*C32*B28)^0.5

d. Still assume that you create a portfolio of 50% stock A and 50% stock B. However, after the initial allocation, you do not rebalance the portfolio at all. What is the holding period return over the 10 weeks for the portfolio?

	A	B	C	D	E
...					
31	Portfolio	Stock A	Stock B		
32	Weights	0.5	0.5		
...					
53	Unrebalanced portfolio				
54	Week	$ stock A	$ stock B	$ total	
55	0	0.5	0.5	1	=B55+C55
56	1	0.5082	0.4894	0.9977	
57	2	0.5458	0.4969	1.0427	
58	3	0.5377	0.5149	1.0526	
59	4	0.5101	0.5246	1.0347	
60	5	0.5396	0.5111	1.0507	
61	6	0.5623	0.5247	1.0871	
62	7	0.5082	0.5136	1.0219	
63	8	0.5124	0.5315	1.0439	
64	9	0.5237	0.5593	1.083	
65	10	0.5387	0.5526	1.0913	
66		=B64*(1+B24)	=C64*(1+C24)		
67					
68	HPR	0.0913			=D65/D55-1
69	Alternative	**0.0913**			=B32*B25+C32*C25

e. What is the standard deviation of weekly returns for such a portfolio if you do not rebalance at all?

	A	B	C	D	E	F
...						
53	Unrebalanced portfolio					
54	Week	$ stock A	$ stock B	$ total	r	
55	0	0.5	0.5	1		
56	1	0.5082	0.4894	0.9977	-0.002348	=D56/D55-1
57	2	0.5458	0.4969	1.0427	0.04513	
58	3	0.5377	0.5149	1.0526	0.009529	
59	4	0.5101	0.5246	1.0347	-0.017025	
60	5	0.5396	0.5111	1.0507	0.015486	
61	6	0.5623	0.5247	1.0871	0.03458	
62	7	0.5082	0.5136	1.0219	-0.05998	
63	8	0.5124	0.5315	1.0439	0.02159	

64	9	0.5237	0.5593	1.083	0.03747	
65	10	0.5387	0.5526	1.0913	0.007634	
...						
70	Average	0.009208				=AVERAGE(E56:E65)
71	Variance	0.0009486				=VAR.S(E56:E65)
72	St. Dev.	**0.0308**				=STDEV.S(E56:E65)

23. Go to Yahoo Finance (https://finance.yahoo.com/) and download weekly stock price data for Target and Walmart for all of 2015, 2016 and 2017.

   a. Use the adjusted close prices to calculate weekly returns. What was the arithmetic average weekly return for Walmart?

   The adjusted close prices are adjusted for dividend payments and stock splits. Therefore, it is likely that the adjusted close prices that you downloaded are different from the ones listed below. However, the returns will be the same.

	A	B	C	D
1	Date	Target	Walmart	
2	1/1/15	67.63	80.20	
3	1/8/15	65.48	78.40	
4	1/15/15	65.15	78.43	
...				
156	12/14/17	62.91	96.99	
157	12/21/17	63.56	97.49	
158	12/28/17	63.66	96.99	
159				
160	**Returns**			
161	Date	Target	Walmart	Formula
162	1/1/15	-0.032	-0.022	=C3/C2-1
163	1/8/15	-0.005	0.000	
164	1/15/15	0.004	0.002	
...				
316	12/14/17	0.010	0.005	
317	12/21/17	0.002	-0.005	
318	12/28/17			
319				
320	Average	0.000188	**0.001502**	=AVERAGE(C162:C317)

   b. What was the standard deviation for Walmart?

	A	B	C	D
...				
321	St.Dev.	0.03379	**0.02396**	=STDEV.S(C162:C317)

   c. What was the covariance of returns?

	A	B	C	D
...				
322	Covar.	**0.0003261**		=COVARIANCE.S(B162:B317,C162:C317)

d. What was the correlation coefficient?

	A	B	C	D
...				
323	Correl.	**0.4028**		=CORREL(B162:B317,C162:C317)

24. Your client is obsessed with technology stocks and wants to invest only in Apple, Google (GOOGL) and Microsoft stock. Download weekly price data for those stocks from Yahoo Finance (http://finance.yahoo.com/) for 2012. For each stock, you should have one tab (named something like Apple, Google, MSFT) that looks like this (Apple example):

	A	B	C	D	E	F	G
1	Date	Open	High	Low	Close	Adj. Close	Volume
2	1/1/12	58.49	60.39	58.43	60.34	40.55	287951300
...							
54	12/30/12	72.93	76.48	72.71	76.02	54.39	164873100

Depending on the date you downloaded the data, your adjusted close prices may be different, but that is irrelevant.

Hint: Select cell B2 and choose Window->Freeze Panes to always keep the date and header row in view.

a. Add a column labeled "Weekly return" to each tab and calculate the total return for each week, using the adjusted close. What was the return for Google from 12/23/12 to 12/30/12?

Hint: If you select all 3 tabs at once by holding control/command before clicking the tab name and add the required formulas to one tab, the changes will appear on all 3 tabs. Don't forget to unselect the 3 tabs by clicking on a different tab when you're done, or you may accidentally overwrite content on other tabs.

The formula in H2 should say =G2/G3-1. Copy and paste the formula all the way down to H53.

	A	...	F	...	H	I
1	Date		Adj. Close		Weekly return	
2	1/1/12		322.91			
3	1/8/12		310.48		-3.85%	=F3/F2-1
...						
53	12/23/12		347.74		-2.18%	=F53/F52-1
54	12/30/12		351.40		1.05%	=F54/F53-1

The weekly return from 12/23 to 12/30 is 351.4/347.74 - 1 = **0.010525**.

Depending on the date you downloaded the data, your adjusted close prices may be different, but the return must be the same.

b. On a new tab, create the covariance matrix using Excel's COVARIANCE.S() formula. What is the weekly covariance of returns for Google and Microsoft based on the historical data (report 7 digits)?

For the covariance between Google and Microsoft, use
= COVARIANCE.S(Google!$H$3:$H$54,MSFT!$H$3:$H$54)
= **0.00015812**

	...	C	D	E	F

1			Apple	Google	MSFT
2		Apple	0.0016435	0.0005068	0.0003228
3		Google	0.0005068	0.0008858	0.0001581
4		MSFT	0.0003228	0.0001581	0.0007587

Note that the diagonal terms, such as Apple-Apple, are actually the variances of returns.

c. On the same tab, create the annualized covariance matrix. Covariance, just like variance, is additive: the covariance of 2-weekly returns is twice the covariance of weekly returns.

What is the annualized covariance of returns for Apple and Google?

We multiply all covariance terms by 52, the number of weeks per year.

For the covariance between Apple and Google in cell E7:
= E2 * $C$6

	...	C	D	E	F
...					
6		52	Apple	Google	MSFT
7		Apple	0.0854611	**0.02635**	0.0167863
8		Google	0.0263519	0.0460602	0.0082221
9		MSFT	0.0167863	0.0082221	0.0394529

d. Create the border-multiplied covariance matrix. What is the variance of the equally-weighted portfolio?

The weights in the vertical and horizontal border must be symmetric and linked. Enter =C13 in cell D12, =C14 in cell E12, and =C15 in cell F12.

Each cell in the border-multiplied covariance matrix is the product of the respective covariance from the covariance matrix and the respective weights. For example, the formula in cell D13 is:
= D7*$C13*D$12

Portfolio variance:
$\sigma_P^2 = \Sigma_i \Sigma_j \, w_i w_j \, \mathrm{cov}(r_i, r_j)$

The portfolio variance is just the sum of those nine cells.

	A	B	C	D	E	F
...						
11				Apple	Google	MSFT
12			Weights	33.33%	33.33%	33.33%
13	Apple		33.33%	0.0094957	0.0029280	0.0018651
14	Google		33.33%	0.0029280	0.0051178	0.0009136
15	MSFT		33.33%	0.0018651	0.0009136	0.0043837
16			100%	= D9*$C15*D$12		
17			=SUM(C13:C15)			
18	σ_P^2	**0.03041**	=SUM(D13:F15)			

e. You did extensive security research and came up with the following annual expected returns:

$E(r_{Apple}) = 0.07$
$E(r_{Google}) = 0.06$

$E(r_{MSFT}) = 0.05$

What is the expected return of the equally-weighted portfolio? Use Excel's SUMPRODUCT() function.

The portfolio expected return equals the weighted average exptected return of the individual securities. The SUMPRODUCT() function is ideal for this purpose:

=SUMPRODUCT(B13:B15, C13:C15) is the same as typing
=B13*C13 + B14*C14 + B15*C15

	A	B	C	D	E	F
...						
11				Apple	Google	MSFT
12		E(r)	Weights	33.33%	33.33%	33.33%
13	Apple	0.07	33.33%	0.0094957	0.0029280	0.0018651
14	Google	0.06	33.33%	0.0029280	0.0051178	0.0009136
15	MSFT	0.05	33.33%	0.0018651	0.0009136	0.0043837
16			100%	= D9*$C15*D$12		
17			=SUM(C13:C15)			
18	σ_P^2	0.03041	=SUM(D13:F15)			
19	E(r)	**0.06**	=SUMPRODUCT(B13:B15, C13:C15)			

f. The yield on treasury bills is 2%. What is the Sharpe ratio of the equally-weighted portfolio?

The Sharpe ratio equals the ratio of portfolio excess return over portfolio standard deviation.

	A	B	C	D	E	F
...						
18	σ_P^2	0.03041	=SUM(D13:F15)			
19	$E(r_P)$	0.06	=SUMPRODUCT(B13:B15, C13:C15)			
20	rf	0.02				
21	S_P	**0.2294**	=(B19-B20)/B18^0.5			

g. Determine the optimal risky portfolio (ORP) using Excel's Solver tool. What is the Sharpe ratio of the portfolio?

Click on Tools->Solver, then make the following choices:

- Select the cell with the Sharpe ratio, B21, as the objective
- Set the goal: To maximize the objective
- By changing the weights in the range C13:C15
- Add a constraint that the weights must sum to 1: C16=1
- Click on "Solve"
- Choose to keep the Solver solution

The spreadsheet should now look like this:

	A	B	C	D	E	F
...						
11				Apple	Google	MSFT
12		E(r)	Weights	**21.1%**	**43.0%**	**35.9%**
13	Apple	0.07	**21.1%**	0.0038096	0.0023922	0.0012720

14	Google	0.06	**43.0%**	0.0023922	0.0085151	0.0012688
15	MSFT	0.05	**35.9%**	0.0012720	0.0012688	0.0050820
16			100%			
17						
18	σ_P^2	0.02727	=SUM(D13:F15)			
19	$E(r_P)$	0.05852	=SUMPRODUCT(B13:B15, C13:C15)			
20	rf	0.02				
21	S_P	**0.2333**	=(B19-B20)/B18^0.5			

h. You have $20,000 to invest, and want to achieve an expected return of 5% on the complete portfolio, including treasury bills. How much money should you put into Apple stock?

The complete portfolio C consists of the risky portfolio P and Treasury bills. Its expected return is the weighted average of the returns of both components:

`E(r_C) = w E(r_P) + (1-w) r_f`

⇔ `w = (E(r_C) - r_f) / (E(r_P) - r_f)= (.05 - 0.02) / (0.05852 - 0.02) = 0.7788`

We have to put 77.88% of funds into the ORP, or $20,000 * 0.7788 = $15,576.

Finally, the weight on Apple in the ORP is 0.2111. Therefore, we buy Apple stock worth $15,576 * 0.2111 = $**3,289**.

25. The return statistics for two stocks and the risk-free asset, Treasury bills, are given below:

	A	B	C	D
1		Stock A	Stock B	T-bills
2	Expected return	0.099	0.065	0.02
3	Variance	0.0961	0.0729	
4	Standard deviation	0.31	0.27	
5	Covariance	0.02511		

a. What is the Sharpe ratio of the optimal risky portfolio?

Use Excel's Solver tool to:

- Set objective: B12 (the Sharpe ratio)
- To: Max (maximize the Sharpe ratio)
- By changing variable cells: B8 (the weight on stock A)

	A	B	C	D
1		Stock A	Stock B	T-bills
2	Expected return	0.099	0.065	0.02
3	Variance	0.0961	0.0729	
4	Standard deviation	0.31	0.27	
5	Covariance	0.02511		
6				
7	**Optimal risky portfolio**			
8	Weights	0.6642	0.3358	=1-B8

9	Expected return	0.08758		=B8*B2+C8*C2
10	Variance	0.06181		=B8^2*B3+C8^2*C3+2*B8*C8*B5
11	Standard deviation	0.2486		=B10^0.5
12	Sharpe ratio	**0.2718**		=(B9-D2)/B11

b. What is the standard deviation of a portfolio on the efficient frontier with an expected return of 9.9%?

Since stock A is on the efficient frontier and has an expected return of 9.9%, the standard deviation of a portfolio on the efficient frontier with such an expected return is the standard deviation of stock A, or **0.31**.

c. What is the standard deviation of the portfolio on the capital market line with an expected return of 9.9%?

	A	B	C	D
1		Stock A	Stock B	T-bills
2	Expected return	0.099	0.065	0.02
3	Variance	0.0961	0.0729	
4	Standard deviation	0.31	0.27	
5	Covariance	0.02511		
6				
7	**Optimal risky portfolio**			
8	Weights	0.6642	0.3358	=1-B8
9	Expected return	0.08758		=B8*B2+C8*C2
10	Variance	0.06181		=B8^2*B3+C8^2*C3+2*B8*C8*B5
11	Standard deviation	0.2486		=B10^0.5
12	Sharpe ratio	0.2718		=(B9-D2)/B11
13				
14	**Complete portfolio**			
15	Weight on ORP	1.169		=(B2-D2)/(B9-D2)
16	Expected return	0.099		=B15*B9+(1-B15)*D2
17	Standard deviation	**0.2906**		=B15*B11

The standard deviation of the portfolio on the capital market line is less than the standard deviation of a portfolio on the efficient frontier with the same expected return, showing the benefit of combining the optimal risky portfolio with the risk-free asset.